

```

Data seeds;
  length variety $12.;
  input block @;
  do rate = 6, 12, 18, 24;
    input @;
    do variety="Legend", "OAC Minto", "Vernal";
      input yield @;
      output;
    end;
  end;
  datalines;
1 8.6 9.5 7.5 9.6 9.6 8.9 11.5 9.2 7.5 9.3 9.9 8.0
2 9.1 9.0 8.0 10.7 11.7 8.9 12.1 12.1 8.9 12.1 10.3 8.1
3 9.7 9.4 9.9 10.0 11.6 7.5 11.6 10.5 8.4 10.8 10.1 7.9
;
Run;

Proc print data=seeds;
Run;

/* Step 1 : ANOVA and means */
Proc glimmix data=seeds;
  class block variety rate;
  model yield = variety|rate;
  random block;
  lsmeans variety|rate;
  covtest "test if block=0" 0 .;
Run;

/* Step 2: Regression partitions */
Proc glimmix data=seeds;
  x=rate;
  class block variety rate;
  model yield = variety x x*x x*rate x*variety x*x*variety x*variety*rate / htype=1;
  random block;
Run;

/* Step 3: Regression coefficients */
Proc glimmix data=seeds;
  x = rate;
  class block variety rate;
  model yield = variety x*variety x*x*variety / htype = 1 noint solution;
  random block;
  parms (0.2134) (0.5427) / noiter;

```

```

Run;

/* Step 4: Regression comparisons */

Proc glimmix data=seeds;
  x = rate;
  class block variety rate;
  model yield = variety x*variety x*x*variety / htype = 1 noint solution;
  random block;
  parms (0.2134) (0.5427) / hold=1,2;

/* Comparing the coefficients - Linear effects */
contrast "Linear coefficients for Legend vs OAC Minto" x*variety -1 1 0;
contrast "Linear coefficients for Legend vs Vernal" x*variety -1 0 1;
contrast "Linear coefficients for OAC Minto vs Vernal" x*variety 0 -1 1;

/* Comparing the coefficients - Quadratic effects */
contrast "Quadratic coefficients for Legend vs OAC Minto" x*x*variety -1 1 0;
contrast "Quadratic coefficients for Legend vs Vernal" x*x*variety -1 0 1;
contrast "Quadratic coefficients for OAC Minto vs Vernal" x*x*variety 0 -1 1;

/* Comparing the coefficients - Quadratic effects */
contrast "Regression for Legend vs OAC Minto" x*variety -1 1 0, x*x*variety -1 1 0;
contrast "Regression for Legend vs Vernal" x*variety -1 0 1, x*x*variety -1 0 1;
contrast "Regression for OAC Minto vs Vernal" x*variety 0 -1 1, x*x*variety 0 -1 1;

/* Comparing the coefficients - Quadratic effects */
contrast "Intercept and regression for Legend vs OAC Minto" variety -1 1 0, x*variety -1 1 0, x*x*variety -1 1
0;
contrast "Intercept and regression for Legend vs Vernal" variety -1 0 1, x*variety -1 0 1, x*x*variety -1 0 1;
contrast "Intercept and regression for OAC Minto vs Vernal" variety 0 -1 1, x*variety 0 -1 1, x*x*variety 0 -1
1;
Run;

/* Visualizing the Regression Lines for the 3 varieties*/
Proc glimmix data=seeds;
  x = rate;
  class block variety rate;
  model yield = variety x*variety x*x*variety / htype = 1 noint solution;
  random block;
  parms (0.2134) (0.5427) / noiter;
  output out=viz_seeds pred=pred;
Run;

```

```

/* Calculating means of yield and predicted yield */
Proc means data=viz_seeds nway;
  class variety rate;
  var yield pred;
  output out=mviz_seeds mean=myield mpred stderr=seyield sepred;
Run;

/* Creating the error bars */
Data mviz_seeds (drop= _TYPE_ _FREQ_);
  set mviz_seeds;
  low = myield - seyield;
  up = myield + seyield;
Run;

/* Visualizing the chart */
Proc sgplot data=mviz_seeds;
  scatter x=rate y=myield / group=variety yerrorlower= low yerrorupper = up;
  series x=rate y=mpred / group=variety lineattrs=(pattern=solid);
  xaxis label = "Seeding Rate (kg/ha)";
  yaxis label = "Herbage Yield (T/ha)";
  keylegend / title = "Variety" location=outside position=right;
  title "Relationship between seeding rate and first production year herbage yield of three varieties of alfalfa
at Elora, Ontario, 1991";
Run;

```